Two Dimensional Treatment of the Magnetic Field due to a circular coil (ring of current)

If we consider the $\vec{B}$ Field created by a ring of current oriented normal to the z axis, we expect that planes passing through the z axis at various θ would exhibit the same $\vec{B}$ Field plots ($\vec{B}$ is symmetric with respect to rotations in the θ direction). Thus, a 2D representation of $\vec{B}$ in any of these planes is sufficient to describe it's behavior.

Here we show the current loop normal to the x-z plane, flowing around the z axis and a possible representation of the $\vec{B}$ field streamlines in the x-z plane. If you can imagine other planes passing through the z axis, similar $\vec{B}$ field streamlines would be expected to occur on these planes due to the θ symmetry. Realize that the ring is actually in the x-y plane (the blue "ring" is misleading -- it's intersection with the x-z plane would just be two points on the x axis (corresponding to the points {R, 0, 0} and {-R,0,0}).

After some interpretation, we will extend the problem to 3 Dimensions. **(you can relax; we do the work).**



Using the Biot-Savart Law, we wish to derive the appropriate $\vec{B}$[x,z]. We will use an approach presented by Phil Duxbury (Michigan State U.) in this notebook: http://computation.pa.msu.edu/-phy201/worksheet8nb/.

(a) We know that Biot-Savart looks like this: $\vec{B} = \int \frac{\mu_o I}{4\pi} \frac{\vec{dl} \times \vec{\mathbb{R}}}{\mathbb{R}^3}$ where $\vec{\mathbb{R}} = \vec{r}$ring - $\vec{r}$fieldpoint.

$\vec{r}$ring is the radius vector of the differential element of the ring $\vec{\text{dl}}$ over which we are integrating.

$\vec{r}$fieldpoint is a point in the plane of interest passing through the z axis and cutting the ring similar to the plane shown in the figure above. (This is the x-z plane and therefore $\vec{r}$fieldpoint = {x, 0, z}).

Enter below the code in M to define $\vec{r}$ring , $\vec{r}$fieldpoint, $\vec{\mathbb{R}}$, and the argument of the Biot-Savart integral: $\left(\frac{\mu_o I}{4\pi}\,\frac{\vec{\text{dl}}\times\vec{\mathbb{R}}}{\mathbb{R}^3}\right)$.

HUGE HELPERS: Letting Ro represent the (constant) radius of the ring, the vector rring = Ro {Cos[theta], Sin[theta], 0} and the vector dl = {0,0,1} × rring (I am leaving off the arrows above $\vec{\text{dl}}$ and $\vec{r}$ring.)

Be sure you verify these 'helpers'. They are written in Cartesian Coordinates and theta is an angular parameter that takes you around the ring. It will be our variable of integration.

Now enter the code:

```
(*  INPUT CELL  *)
ClearAll["Global`*"]
```

(b) Hanging firmly onto your hat, integrate these three terms using M to obtain expressions for B in the x-z plane. (M fearlessly attacks these integrals and will generate solutions involving Elliptical Integrals).

Important note: It took M too long to do the definite integrals, so I did the indefinites first, then evaluated the limits (0 to 2π); worked like a champ.

I will first do each component separately, then combine into (hopefully) a 2D vector with x and z components only. The integrals are over theta from 0 to 2 π. M does the indefinite integral extremely fast whereas the definite integral is very slow to emerge. Therefore I do the indefinite and then evaluate at the endpoints (0 and 2 π)

In[61]:=
```
(*  II = 1; Ro =1; μo = 1;   *)
dB[[1]] (*  note that it is a function of theta *)
Bx[theta_] = Integrate [dB[[1]], theta] (* dB[[1]] is the x component of dB;
this is the indefinite integral over theta *)
Bxx = Bx[2 π] - Bx[0] // Simplify
(*  Here is the final definite integral for the x component  *)

dB[[2]] (*  note that it is a function of theta *)
By[theta_] = Integrate [dB[[2]], theta]
(* dB[[2]] is the y component of dB;  this is the indefinite integral *)
Byy = By[2 π] - By[0] // Simplify
(*  Here is the final definite integral for the y component  *)

dB[[3]] (*  note that it is a function of theta *)
Bz[theta_] = Integrate [dB[[3]], theta]
(* dB[[3]] is the z component of dB;  this is the indefinite integral *)
Bzz = Bz[2 π] - Bz[0] // Simplify
  (*  Here is the final definite integral for the z component  *)
```

Messy! but M can evaluate the Elliptical Functions to very high precision, very quickly. NOTE that Byy = 0 so the B vectors lie in the plane chosen (the x-z plane).

(c) From earlier work we know that $\vec{B}$ along the symmetry axis (the z axis) $\vec{B}$zaxis = $\frac{\text{II Ro}^2\,\mu\text{o}}{2\,(\text{Ro}^2+z^2)^{3/2}}\hat{z}$.

Check to see if the above mess reduces to this result.   Recommend you use a Limit statement like this: Limit[{Bxx, Bzz}, x→ 0]

(d) From earlier work we know that $\vec{B}$ at the center of the current loop (the origin) = $\frac{\text{II}\,\mu\text{o}}{2\ \text{Ro}}\hat{z}$. Check to see if the above reduces to this result. Recommend you use a Limit statement like this: Limit[{Bxx, Bzz}, {x→ 0,z→ 0},Assumptions→ Ro >0]

(e) Generate a VectorPlot and StreamPlot for B in the x-z plan using the results from the integrations above. Be sure you understand where the current loop is sitting in space. I used these values for the parameters: II = 1; Ro = 1; *μ*o = 1;

Where is the current loop?: It intersects the x-z plane inside the 'circles' (at {0, ± 1}) and lies normal to this plane. It's symmetry axis is the z axis.

(f) Explore the Divergence of B and comment using M’s Div function.

Doesn’t look like zero; We can plot of Div[$\vec{B}$] we see that it is <u>not</u> zero!

```
II = 1; Ro = 1; μo = 1;
Plot3D[divvy, {x, -2, 2}, {z, -2, 2}, PlotRange → {-1.2, 1.2}]
```

What is wrong??? This type of behavior arises all the time when you reduce the problem to 2 (or 1) dimension(s).

IN ONE D: Recall that $\vec{B}$ along the symmetry axis of the ring (examined above) is $\vec{B}[z] = \frac{\text{II Ro}^2\,\mu o}{2\left(\text{Ro}^2+z^2\right)^{3/2}}\hat{z}$. Since this is a function of z only (and $\vec{B}$ points in the z direction only) we would assume that Div[$\vec{B}$[z]] is given by: D[B[z], z] ($\vec{B}$[z] is a vector; B[z] is the functional form of the length of $\vec{B}$[z]).

DO IT:

$$D\left[\frac{\text{II Ro}^2\,\mu o}{2\left(\text{Ro}^2+z^2\right)^{3/2}}, z\right]$$

Here (in the 1D case) and in the 2D case we are ignoring the fact that there are components of B in the x and/or y direction adjacent to the points where we are evaluating the divergence. E.g., for $\vec{B}$ along the symmetry axis (z), even though $\vec{B}$ is only in the z direction, if we are not careful we might conclude that Div B ≠ 0 (which means there are magnetic monopoles -- experimentally, so far, their ain’t any!).

Reminder: Form of Div in Cartesian Coordinates.

Take the “dummy” vector function BB = {BBx[x, y, z], BBy[x, y, z], BBz[x, y, z]} and take apply Div:

```
In[4]:= Div[{BBx[x, y, z], BBy[x, y, z], BBz[x, y, z]}, {x, y, z}]
```

Converting back to $\vec{B}$, The term(s) we are missing in the divergence is $By^{(0,1,0)}$[x, y, z] which = D[By[x,y,z], y] and $Bz^{(0,0,1)}$[x, y, z] which = D[Bz[x,y,z], z]

We expect Div[$\vec{B}$] = 0 when we have all three components of $\vec{B}$ correct.

SO, let’s try to solve the 3D problem for the ring of current. This is on us.

We start just like above but we let the field point be in 3D: rfieldpoint = {x, y, z};
Again, theta is being treated like a parameter which takes us around the current loop (it actually is the polar angle in the x-y plane); it’s the variable of integration.

In[21]:=
```
ClearAll["Global`*"]
rring = Ro {Cos[theta], Sin[theta], 0} ;
(*parametric version of the source points on the ring *)
rfieldpoint = {x, y, z};
(* we are examining B NOW in 3D so rfieldpoint = {x,y,z} *)
dl = {0, 0, 1} × rring
(*  NOTE:  the smaller x:  ×  is for the vector cross-product *)
ℝ = rfieldpoint - rring
ℝmag = √(ℝ.ℝ)   (* ℝmag is the length of the VECTOR ℝ  *)
dB = (μo II)/(4 π) (dl × ℝ)/(ℝmag^3)  // Simplify
```

M does better on the calculus if we convert the dB we have in Cartesian coordinates to dB = dBcyl in Cylindrical Coordinates and then integrate. We use M's TransformedField to do this coordinate transformation:

In[28]:=
```
dBcyl = TransformedField["Cartesian" → "Cylindrical", dB, {x, y, z} → {s, ϕ, ξ}] //
  Simplify (* a 3D Vector *)
```

In[32]:=
```
(*  NOW  integrate to find the three components of B in cylindrical
 coordinates  Bss, Bϕϕ, and Bξξ   where ξ is the Greek z  *)
$Assumptions = {II, Ro, μo, theta, s, ϕ, ξ} ∈ Reals && 0 ≤ {theta, ϕ} ≤ 2 π;
dBcyl[[1]] (*  note that it is a function of theta *)
Bss[theta_] = Integrate[dBcyl[[1]], theta] // Simplify ;
(*  this is the indefinite integral *)
Bss[s_, ϕ_, ξ_] = (Bss[2 π] - Bss[0])  //
  Simplify  (*  the definite integral over 0 ≤ theta ≤ 2  π *)
```

In[35]:=
```
dBcyl[[2]] (*  note that it is a function of theta *)
Bϕϕ[theta_] = Integrate[dBcyl[[2]], theta] // Simplify ;
(*  this is the indefinite integral *)
Bϕϕ[s_, ϕ_, ξ_] = (Bϕϕ[2 π] - Bϕϕ[0])  //
  Simplify  (*  the definite integral over 0 ≤ theta ≤ 2  π *)
```

We should get B$\phi\phi$ = 0 due to the cylindrical symmetry of the current distribution.

In[38]:=
```
dBcyl[[3]] (*  note that it is a function of theta *)
Bξξ[theta_] = Integrate[dBcyl[[3]], theta] // Simplify ;
(*  this is the indefinite integral *)
Bξξ[s_, ϕ_, ξ_] = (Bξξ[2 π] - Bξξ[0])  //
  Simplify  (*  the definite integral over 0 ≤ theta ≤ 2  π *)
```

Div[B] = 0 in (3D) cylindrical coordinates. [Hurray!]   No Monopoles.

In[45]:=
```
BB[s_, ϕ_, ξ_] = {Bss[s, ϕ, ξ], Bϕϕ[s, ϕ, ξ], Bξξ[s, ϕ, ξ]};
Div[BB[s, ϕ, ξ] , {s, ϕ, ξ}, "Cylindrical"] // FullSimplify
```

Expect $\vec{B}$ along the symmetry axis (the z axis) $\vec{B}$zaxis to equal: $\frac{\text{II Ro}^2\,\mu\text{o}}{2\,(\text{Ro}^2+z^2)^{3/2}}\hat{z}$. I.e., Bss and B$\phi\phi$ = 0

(we showed above that B$\phi\phi$ = 0 for all points in space; performing the following shows that Bss = 0 for s = 0.

I.e., we get: $\vec{B}$zaxis = $\left\{0,\ 0,\ \frac{\text{II Ro}^2\,\mu\text{o}}{2\,(\text{Ro}^2+\xi^2)^{3/2}}\right\}$.)

In[52]:=
```
Limit[BB[s, ϕ, ξ], s → 0] (* if you get a Warning about assumptions, ignore *)
```

We can find B at the center of the ring:

In[53]:=
```
Limit[%, ξ → 0, Assumptions → Ro > 0] // Simplify
```

It is redundant but we can quickly show that Bss = 0 for s = 0

In[381]:=
```
Limit[Bss[s, ϕ, ξ], s → 0] (* if you get a Warning about assumptions, ignore *)
```

Plot B in cylindrical coordinates for case of ϕ = 0. This will be like a Cartesian plot where s -> x and ζ -> z. Note that s is always positive so we can only represent “half” the x - z plane:

In[382]:=
```
Clear[II, Ro, μo, s, ϕ, ξ]
II = 1; Ro = 1; μo = 1;
BB2D[s_, ξ_] = {BB[s, ϕ, ξ][[1]], BB[s, ϕ, ξ][[3]]} /. ϕ → 0
StreamPlot[BB2D[s, ξ] , {s, 0.01, 2}, {ξ, -2, 2}, AspectRatio → Automatic]
```

The current loop is coming out/going into the paper at the point {1,0}, inside the ‘circle’. For the Field direction shown by the streamlines, the current is going into the paper.

Transform to Cartesian Coordinates - This is the full 3D solution for BB in {x,y,z} coordinates. The functions can be simplified: **This article derives fairly simple expressions for Cartesian, Spherical, and Cylindrical coordinates in 3D**

In[54]:=
```
Clear[II, Ro, μo]
Bcart3D[x_, y_, z_] = TransformedField[
   "Cylindrical" → "Cartesian", BB[s, ϕ, ξ], {s, ϕ, ξ} → {x, y, z}] // Simplify
```

Div[B] works in (3D) Cartesian coordinates just fine:

In[56]:=
```
Div[Bcart3D[x, y, z] , {x, y, z}, "Cartesian"] // Simplify
```

Bcard3D evaluated along the z axis is, as before [we should get: $\left\{0, 0, \frac{\text{II Ro}^2 \mu\text{o}}{2 \left(\text{Ro}^2+z^2\right)^{3/2}}\right\}$ = $\frac{\text{II Ro}^2 \mu\text{o}}{2 \left(\text{Ro}^2+z^2\right)^{3/2}} \hat{z}$]

Letting M actually Do It (in Cartesian Coordinates):

In[57]:=
```
Limit[Bcart3D[x, y, z], {x → 0, y → 0}]
(*  THIS TAKES A LONG TIME -on an iMac, ~2 minutes  *)
```

We can plot Bcard3D either in 3D: (Blind use of VectorPlot3D -- it’s not pretty)

In[58]:=
```
II = 1; Ro = 1; μo = 1;
VectorPlot3D[Bcart3D[x, y, z], {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
(* 3D:  this graph is a mess -- much playing around might help  *)
```

We can plot Bcard3D in 2D as we did above. we set y -> 0.

```
(*  2D Plots:  Just set y → 0;
Results will look just like the 2D results obtained above *)
Bxx = Bcart3D[x, y, z][[1]] /. y → 0;  (* remove ;
if you want to see functional forms  *)
Bzz = Bcart3D[x, y, z][[3]] /. y → 0;
VectorPlot[{Bxx, Bzz}, {x, -2, 2}, {z, -2, 2}, VectorPoints → 22,
 VectorScale → {Medium, Scaled[0.7]}, VectorStyle → Yellow,
 FrameStyle → White, Background → Black, FrameLabel → {x, z}]
StreamPlot[{Bxx, Bzz}, {x, -2, 2}, {z, -2, 2},
 StreamPoints → 26, StreamColorFunction → Hue]
```

Exactly as the 2D results obtained above.